
Limiti di MPI
I Ogni processo MPI può accedere soltanto alla sua

memoria locale
I I dati da condividere devono essere scambiati con

comunicazioni esplicite tra processi (messaggi)
I È onere del programmatore progettare ed implementare lo

scambio dei dati tra i processi
I Non è possibile adottare una strategia di parallelizzazione

incrementale
I Deve essere implementata la struttura di comunicazione

dell’intero programma
I Le comunicazioni hanno un costo
I È difficile avere il programma seriale e quello MPI in un’unica

versione del codice
I Sono necessarie variabili aggiuntive
I È necessario gestire la corrispondenza tra variabili locali e

struttura dati globale
I La gestione del sorgente è più complessa

Cos’è OpenMP

I “Open specifications for Multi Processing”
(http://www.openmp.org/)

I Modello di programmazione parallela per architetture a
memoria condivisa (shared memory )

I I “lavoratori” che compiono il lavoro in parallelo (thread)
“cooperano” attraverso la memoria condivisa

I Accessi alla memoria invece di messaggi espliciti
I Modello di parallelizzazione “locale” del codice seriale

I Permette la parallelizzazione incrementale
I Riduce il tempo di startup della parallelizzazione
I Un unico sorgente per il seriale e il parallelo!



La storia di OpenMP
I Nato come unificazione di soluzioni proprietarie
I Il passato:

I ottobre 1997 - versione 1.0 Fortran
I ottobre 1998 - versione 1.0 C/C++
I novembre 1999 - versione 1.1 Fortran (interpretazioni)
I novembre 2000 - versione 2.0 Fortran
I marzo 2002 - versione 2.0 C/C++
I maggio 2005 - versione 2.5 unificazione Fortran/C/C++
I maggio 2008 - versione 3.0 (task !)
I Luglio 2011 - versione 3.1 released

I Il presente:
I task final
I atomics e C/C++ min max reduction
I initial support for thread binding
I many clarifications, improvements to examples (100 pages)

I Il futuro:
I novembre 2013 - versione 3.5/4.0

I support for accelerators (GPUs, MIC, FPGA, etc)
I many improvements on tasks and atomics
I custom reduction operations

Architetture Shared Memory

I Visione logica
I La memoria è accessibile da tutti i processori del computer

I Visione fisica
I Uniform Memory Access

I I tempi di accesso in memoria sono
gli stessi per qualsiasi CPU

I Non Uniform Memory Access
I I tempi di accesso in memoria

dipendono dalla posizione della
CPU rispetto alla locazione di
memoria



Processo e thread

I Il processo è un’istanza del
programma in esecuzione

I Alcune informazioni contenute nel
processo:

I Text
I codice macchina

I Data
I variabili globali

I Stack
I variabili locali alle funzioni

I Program Counter (PC)
I puntatore all’istruzione

corrente

Multi-threading

I Il processo contiene diversi flussi
di esecuzione concorrenti (thread)

I Ogni thread ha il suo program
counter (PC)

I Ogni thread ha il suo stack
privato (variabili locali al thread)

I Le istruzioni “puntate” dal PC di
un thread possono accedere:

I Alla memoria globale del
processo (data)

I Allo stack locale del thread



Processori, processi e thread

I I processori sono entità fisiche
composti da uno o più core

I Processi e thread sono entità
logiche

OpenMP: modello fork & join

I Ogni programma OpenMP inizia con un solo thread di
esecuzione che esegue il programma in seriale

I All’inizio di una regione parallela il thread crea un team di
thread composto da se stesso (Master Thread) e da un
insieme di altri thread

I Tutti i membri del team di thread (MT compreso) eseguono in
parallelo il codice contenuto nella regione parallela (modello
SPMD: Single Program Multiple Data)

I Al termine della regione parallela i thread del gruppo
terminano l’esecuzione e solo MT continua l’esecuzione
(seriale) del programma



Direttive

I Sintassi:
I in C/C++:

#pragma omp direttiva
I in Fortran (free format):

!$omp direttiva
I in Fortran (fixed format):

c$omp direttiva

I Annotano un blocco di codice
I Specificano al compilatore come eseguire in parallelo il blocco

di codice

I Il codice seriale “convive” con quello parallelo
I Una compilazione seriale ignora le direttive
I Una compilazione con supporto di OpenMP ne tiene conto

Clausole
I Sintassi:

I direttiva [clausola[ clausola]...]

I Specificano informazioni addizionali alle direttive
I Gestione delle variabili

I Quali sono condivise tra tutti i thread (è il default)
I Quali sono private ad ogni thread
I Come inizializzare quelle private
I Quale è il default

I Controllo dell’esecuzione
I Quanti thread nel team
I Come distribuire il lavoro

I ATTENZIONE: alterano la semantica del codice
I Il codice può essere corretto in seriale ma non in parallelo
I O viceversa
I O cambiare comportamento col numero di thread



Variabili di ambiente e funzioni

I Variabili di ambiente
I OMP_NUM_THREADS: imposta il numero di thread che devono

essere creati in ogni regione parallela
I OMP_DYNAMIC: consente di cambiarne il numero

automaticamente
I ...

I Funzioni
I Determinano/modificano alcuni aspetti dell’ambiente di

esecuzione
I Permettono di gestire accessi a grana fine (lock )
I Misurano il tempo trascorso
I ...ma il codice non è più compilabile in seriale...

I Interfaccia alle funzioni:
I Fortran: use omp_lib
I C: #include <omp.h>

Prodotto matrice matrice
Fortran
program mat_prod

implicit none

integer, parameter :: n=500
integer, parameter :: dp=kind(1.d0)
real(dp), dimension(n,n) :: a, b, c
real :: maxd
integer :: i, j, k

call random_number(a)
call random_number(b)
c = 0.d0

!$omp parallel
!$omp do

do j=1, n
do k=1, n

do i=1, n
c(i,j) = c(i,j) + a(i,k)*b(k,j)

end do
end do

end do
!$omp end do
!$omp end parallel

!check the result
maxd = maxval(abs(c-matmul(a,b)))
write( *,"(e18.9)" ) maxd

end program mat_prod

C
#include <stdio.h>
#include <stdlib.h>
#include <math.h>

int main(int argc,char **argv) {
const unsigned n=500;
double a[n][n],b[n][n],c[n][n],d[n][n];
double rmax = (double)RAND_MAX;
int i, j, k;
for (i=0; i<n; i++)

for (j=0; j<n; j++) {
a[i][j] = ((double)rand())/rmax;
b[i][j] = ((double)rand())/rmax;
c[i][j] = 0.0;

}
#pragma omp parallel
#pragma omp for private(j,k)

for (i=0; i<n; ++i)
for (k=0; k<n; k++)

for (j=0; j<n; ++j)
c[i][j] += a[i][k]*b[k][j];

// check the result
dgemm_(...);
double maxd = 0.0;
for (i=0; i<n; i++)

for (j=0; j<n; j++)
maxd=fmax(maxd,fabs(c[i][j]-d[j][i]));

printf("%18.9lE\n", maxd);
return 0;

}



costrutto parallel

I È il costrutto fondamentale di OpenMP e serve a creare una
regione parallela

I Un costrutto è l’ambito lessicale a cui si applica una direttiva nel
codice sorgente

I Una regione è l’ambito dinamico dell’esecuzione del codice
contenuto in un costrutto

I Ogni thread in una regione parallela esegue il codice
contenuto e invocato al suo interno

Il costrutto do/for

I Distribuisce le iterazioni del loop tra i thread nel team
I La variabile del loop è resa privata automaticamente
I Ogni thread aggiorna componenti diverse di c

I I loop interni sono eseguiti in modo sequenziale da ogni thread
I Attenzione!

I In Fortran, le variabili di iterazione dei loop sequenziali contenuti
all’interno di un costrutto parallelo sono private per default

I In C no...



Regioni parallele: variabili shared e private

I All’interno di una regione parallela, le variabili del programma
seriale possono essere shared o private

I Variabili shared : ogni thread può leggerle/scriverle
I Variabili private: ogni thread ha una sua copia privata

I È invisibile agli altri thread
I Non è inizializzata

I Per default le variabili sono shared, ma è possibile cambiarlo
con una clausola:

I default(none) impedisce ogni assunzione
I default(private) in Fortran le rende private per default

Forma contratta

Fortran

program mat_prod
implicit none

integer, parameter :: n=500
integer, parameter :: dp=kind(1.d0)
real(dp), dimension(n,n) :: a, b, c
real :: maxd
integer :: i, j, k

call random_number(a)
call random_number(b)
c = 0.d0

!$omp parallel do
do j=1, n

do k=1, n
do i=1, n

c(i,j) = c(i,j) + a(i,k)*b(k,j)
end do

end do
end do

!$omp end parallel do

!check the result
maxd = maxval(abs(c-matmul(a,b)))
write( *,"(e18.9)" ) maxd

end program mat_prod

C

#include <stdio.h>
#include <stdlib.h>
#include <math.h>

int main(int argc,char **argv) {
const unsigned n=500;
double a[n][n],b[n][n],c[n][n],d[n][n];
double rmax = (double)RAND_MAX;
int i, j, k;
for (i=0; i<n; i++)

for (j=0; j<n; j++) {
a[i][j] = ((double)rand())/rmax;
b[i][j] = ((double)rand())/rmax;
c[i][j] = 0.0;

}
#pragma omp parallel for private(j,k)

for (i=0; i<n; ++i)
for (k=0; k<n; k++)

for (j=0; j<n; ++j)
c[i][j] += a[i][k]*b[k][j];

// check the result
dgemm_(...);
double maxd = 0.0;
for (i=0; i<n; i++)

for (j=0; j<n; j++)
maxd=fmax(maxd,fabs(c[i][j]-d[j][i]));

printf("%18.9lE\n", maxd);
return 0;

}



Compilazione ed esecuzione
I I compilatori che supportano OpenMP interpretano le

direttive solo se vengono invocati con un’opzione (switch) di
compilazione (-fopenmp per il compilatore GNU)

I Ad esempio il compilatore Intel
I In C: icc -openmp ...
I In Fortran: ifort -openmp ...

I Il numero di thread di esecuzione si definisce a runtime
impostando la variabile di ambiente OMP_NUM_THREADS. Ad
esempio per girare un programma OpenMP con 4 thread :

I In csh/tcsh: setenv OMP_NUM_THREADS 4
I In sh/bash: export OMP_NUM_THREADS=4

I Parallelizziamo e compiliamo il programma prodottoMM
I ifort -openmp prodottoMM.f90 -o prodottoMMf
I icc -openmp prodottoMM.c -o prodottoMMc

I Eseguiamolo variando il numero di thread :
I setenv OMP_NUM_THREADS n
I prodottoMMf;prodottoMMc

Un comando utile: htop

Il calcolo di π

I Il valore di π può essere calcolato tramite∫ 1

0

4
1 + x2 dx = 4 arctan x

∣∣1
0 = π

I E approssimato con il metodo di Riemann dalla serie

N∑
i=1

4h

1 +
[
(i − 1

2)h
]2 con h =

1
N

Paralleliziamo il calcolo di π

Fortran

program pigreco
implicit none

integer, parameter :: li = selected_int_kind(18)
integer(li) :: i
integer(li), parameter :: n=1e8
real(kind(1.d0)) :: dx, sum, x, f, pi

print *, ’number of intervals: ’, n
sum=0.d0
dx=1.d0/n
do i=1,n
x=dx*(i-0.5d0)
f=4.d0/(1.d0+x*x)
sum=sum+f

end do
pi=dx*sum
print ’(a13,2x,f30.25)’,’ Computed PI =’, pi
...

end program



Paralleliziamo il calcolo di π
C

#include <stdio.h>
#define N 100000000

int main(int argc, char **argv)
{

long int i, n = N;
double x, dx, f, sum, pi;

printf("number of intervals: %ld\n", n);
sum = 0.0;
dx = 1.0 / (double) n;
for (i = 1; i <= n; i++) {
x = dx * ((double) (i - 0.5));
f = 4.0 / (1.0 + x*x);
sum = sum + f;

}
pi = dx*sum;
printf("Computed PI %.24f\n", pi);
...
return 0;

}

Parallelizziamo il calcolo di π
Fortran

!$omp parallel do
do i=1,n
x=dx*(i-0.5d0)
f=4.d0/(1.d0+x*x)

sum=sum+f

end do
!$omp end parallel do

pi=dx*sum

C

#pragma omp parallel for
for (i = 1; i <= n; i++) {
x = dx * ((double) (i - 0.5));
f = 4.0 / (1.0 + x*x);

sum = sum + f;
}
pi = dx*sum;



Controllo del Data-Sharing
I Compiliamo e testiamo il programma
I Variando il numero di thread
I Otteniamo risultati differenti ed errati

I Per default, tutte le variabili sono shared
I Tutti i thread modificano x ed f senza coordinarsi
I “Gareggiano” tra loro, è una data race

I È evidente che i valori di x ed f sono specifici ad ogni
iterazione

I Non è necessario condividerle
I La clausola private rende gli elementi della lista “locali” ai

thread
I Ogni thread accede ad una copia privata
I In ingresso ed uscita dal costrutto le variabili private

assumono valori indefiniti

Parallelizziamo il calcolo di π
Fortran

!$omp parallel do private(x,f)
do i=1,n
x=dx*(i-0.5d0)
f=4.d0/(1.d0+x*x)

sum=sum+f

end do
!$omp end parallel do

pi=dx*sum

C

#pragma omp parallel for private(x,f)
for (i = 1; i <= n; i++) {
x = dx * ((double) (i - 0.5));
f = 4.0 / (1.0 + x*x);

sum = sum + f;
}
pi = dx*sum;



Sezioni critiche

I Compiliamo e testiamo il programma
I Variando il numero di thread
I Ed otteniamo ancora risultati differenti ed errati

I C’è un’altra data race
I Tutti i thread leggono e modificano sum senza coordinarsi
I Ma è necessario che sum sia shared per ottenere un risultato

dopo il loop parallelo

I È necessario che i thread aggiornino il contenuto di sum uno
per volta

I Il costrutto critical è un modo di ottenerlo

Parallelizziamo il calcolo di π
Fortran

!$omp parallel do private(x,f)
do i=1,n
x=dx*(i-0.5d0)
f=4.d0/(1.d0+x*x)

!$omp critical
sum=sum+f

!$omp end critical
end do

!$omp end parallel do
pi=dx*sum

C

#pragma omp parallel for private(x,f)
for (i = 1; i <= n; i++) {
x = dx * ((double) (i - 0.5));
f = 4.0 / (1.0 + x*x);

#pragma omp critical
sum = sum + f;

}
pi = dx*sum;



Prestazioni

I Compiliamo e testiamo il programma
I Variando il numero di thread
I E misurando il tempo impiegato

I con il comando time
I o con la funzione omp_get_wtime() (che riporta un valore di

tipo double)
I I risultati ora sono corretti, ma il tempo impiegato è enorme

I Tutti i thread, ad ogni iterazione, sono in contesa per l’accesso
alla sezione critica

I Deve esserci un modo più efficiente

Operazioni collettive: reduction

I reduction(op:list)

I Per ogni variabile nella lista, genera copie locali ai thread
I La copia locale delle variabili contenute in list è inizializzata

in modo opportuno in funzione dell’operatore op scelto (ad
esempio, se op è + le variabili locali sono inizializzate a 0)

I Tutti gli aggiornamenti sono eseguiti sulla copia locale
I E solo alla fine del costrutto viene aggiornata la variabile

originale, con una sezione critica generata automaticamente

I Le operazioni op valide in una reduction sono:
I C: +, *, -, &, |, ^, &&, ||, max, min
I Fortran: +, *, -, .and., .or., .eqv., .neqv.,
max, min, iand, ior, ieor



Parallelizziamo il calcolo di π
Fortran

!$omp parallel do private(x,f) reduction(+:sum)
do i=1,n
x=dx*(i-0.5d0)
f=4.d0/(1.d0+x*x)

sum=sum+f

end do
!$omp end parallel do

pi=dx*sum

C

#pragma omp parallel for private(x,f) reduction(+:sum)
for (i = 1; i <= n; i++) {
x = dx * ((double) (i - 0.5));
f = 4.0 / (1.0 + x*x);

sum = sum + f;
}
pi = dx*sum;

L’equazione di Laplace

I Formula discreta su griglia 2D con condizioni al contorno di
Dirichelet

f (xi+1,j) + f (xi−1,j)− 2f (xi,j)+
f (xi,j+1) + f (xi,j−1)− 2f (xi,j) = 0 ∀xi,j ∈ (a,b)2

f (x) = α(x) ∀xi,j ∈ ∂[a,b]2

I Soluzione iterativa con il metodo di Jacobi
fn+1(xi,j) =

1
4 [ fn(xi+1,j) + fn(xi−1,j)+

fn(xi,j+1) + fn(xi,j−1) ] ∀n > 0
f0(xi,j) = 0 ∀xi,j ∈ (a,b)2

fn(x) = α(x) ∀x ∈ ∂[a,b]2, ∀n ≥ 0

Paralleliziamo il metodo di Jacobi
Fortran
program laplace

integer, parameter :: dp=kind(1.d0)
integer :: n, maxIter, i, j, iter = 0
real (dp), dimension(:,:), pointer :: T, Tnew, Tmp=>null()
real (dp) :: tol, var = 1.d0, top = 100.d0
write( *,* ) ’Enter mesh size, max iter and tol:’
read( *,* ) n, maxIter, tol
allocate (T(0:n+1,0:n+1), Tnew(0:n+1,0:n+1))
T(0:n,0:n) = 0.d0;
T(n+1,1:n) = (/ (i, i=1,n) /) * (top / (n+1))
T(1:n,n+1) = (/ (i, i=1,n) /) * (top / (n+1))
Tnew = T
do while (var > tol .and. iter <= maxIter)
iter = iter + 1; var = 0.d0
do j=1, n
do i=1, n
Tnew(i,j)=0.25d0*(T(i-1,j)+T(i+1,j)+T(i,j-1)+T(i,j+1))
var = max(var,abs( Tnew(i,j) - T(i,j) ))

end do
end do
Tmp =>T; T =>Tnew; Tnew => Tmp;
if (mod(iter,100)==0) write( *,* ) ’ iter, var:’,iter,var

end do
...



Paralleliziamo il metodo di Jacobi
C
int main() {

double *T, *Tnew, *Tmp;
double tol, var=DBL_MAX, top=100.0;
unsigned n, stride, maxIter, i, j, iter = 0;
int itemsread;
printf("Enter mesh size, max iterations and tolerance: ");
itemsread = scanf("%u ,%u ,%lf", &n, &maxIter, &tol);
stride = n+2;
T = calloc(stride*stride, sizeof( *T ));
Tnew = calloc(stride*stride, sizeof( *T ));
setBoundaryConditions(T, Tnew, stride, top);
while(var > tol && iter <= maxIter) {

++iter; var = 0.0;
for (i=1; i<=n; ++i)
for (j=1; j<=n; ++j) {
Tnew[i*stride+j] = 0.25*(T[(i-1)*stride+j]+
T[(i+1)*stride+j]+T[i*stride+(j-1)]+T[i*stride+(j+1)]);
var = MAX(var, ABS(Tnew[i*stride+j] - T[i*stride+j]));
}
Tmp=T; T=Tnew; Tnew=Tmp;
if (iter%100 == 0)

printf("iter: %8u, var=%12.4lE\n",iter,var);
}
...

L’equazione di Laplace: l’algoritmo

I L’array T rappresenta la funzione f (x , y) sui punti della griglia
I T è sovradimensionato in modo da ospitare le condizioni al

contorno
I l’algoritmo calcola solo sui punti interni dell’array

I Ad ogni iterazione, nei punti della griglia si calcola
I la nuova soluzione Tnew(i,j) =
0.25*(T(i-1,j)+T(i+1,j)+T(i,j-1)+T(i,j+1))

I la differenza (var) tra Tnew e la soluzione precedente T

I Il procedimento iterativo termina quando si raggiunge la
convergenza entro la tolleranza oppure si supera il numero
massimo di iterazioni (maxiter)

I Esercizio: parallelizzare il codice seriale



Parallelizzazione incrementale: Fortran

Fortran

do while (var > tol .and. iter <= maxIter)
iter = iter + 1;
var = 0.d0

!$omp parallel do reduction(max:var)
do j=1, n
do i=1, n
Tnew(i,j)=0.25d0*(T(i-1,j)+T(i+1,j)+T(i,j-1)+T(i,j+1))
var = max(var,abs( Tnew(i,j) - T(i,j) ))

end do
end do

!$omp end parallel do
Tmp =>T; T =>Tnew; Tnew => Tmp;
if (mod(iter,100)==0) write( *,* ) ’ iter, var:’,iter,var

end do

Parallelizzazione incrementale: C

C

while(var > tol && iter <= maxIter) {
++iter;
var = 0.0;

#pragma omp parallel
{

double pvar = 0.0;
#pragma omp for private(j)

for (i=1; i<=n; ++i)
for (j=1; j<=n; ++j) {
Tnew[i*stride+j] = 0.25*(T[(i-1)*stride+j]+
T[(i+1)*stride+j]+T[i*stride+(j-1)]+T[i*stride+(j+1)]);

pvar = MAX(pvar, ABS(Tnew[i*stride+j]-T[i*stride+j]));
}

#pragma omp critical
if( pvar > var) var = pvar;

}
Tmp=T; T=Tnew; Tnew=Tmp;
if (iter%100 == 0)
printf("iter: %8u, var=%12.4lE\n",iter,var);

}



Ridurre l’overhead

I Ad ogni iterazione del while viene aperta e chiusa una
regione parallela

I Ed aprire e chiudere una sezione parallela ha un costo
I I thread devono essere creati all’inizio e distrutti alla fine
I Lo stesso per le strutture dati necessarie al runtime per gestirli

e distribuire il lavoro

I Molte implementazioni riducono tali costi con una serie di
trucchi

I Ma in generale è meglio che le sezioni parallele siano più
ampie possibili

I Il che richiede maggiore attenzione nel coordinare i thread
I E nuovi costrutti...

I Procedendo per passi

Parallelizzazione incrementale: Fortran

Fortran

!$omp parallel
do while (var > tol .and. iter <= maxIter)
iter = iter + 1;
var = 0.d0

!$omp do reduction(max:var)
do j=1, n
do i=1, n
Tnew(i,j)=0.25d0*(T(i-1,j)+T(i+1,j)+T(i,j-1)+T(i,j+1))
var = max(var,abs( Tnew(i,j) - T(i,j) ))

end do
end do

!$omp end do
Tmp =>T; T =>Tnew; Tnew => Tmp;
if (mod(iter,100)==0) write( *,* ) ’ iter, var:’,iter,var

end do
!$omp end parallel



Parallelizzazione incrementale: C
C

#pragma omp parallel
{
while(var > tol && iter <= maxIter) {

++iter;
var = 0.0;

double pvar = 0.0;
#pragma omp for private(j)

for (i=1; i<=n; ++i)
for (j=1; j<=n; ++j) {
Tnew[i*stride+j] = 0.25*(T[(i-1)*stride+j]+
T[(i+1)*stride+j]+T[i*stride+(j-1)]+T[i*stride+(j+1)]);

pvar = MAX(pvar , ABS(Tnew[i*stride+j]-T[i*stride+j]));
}

#pragma omp critical
if( pvar > var) var = pvar;

Tmp=T; T=Tnew; Tnew=Tmp;
if (iter%100 == 0)
printf("iter: %8u, var=%12.4lE\n",iter,var);

}
}

Il costrutto single

I Problemi:
I Ora c’è una data race sull’azzeramento di var e l’incremento di
iter

I Lo scambio dei puntatori deve essere fatto da un solo thread
I Di stampa delle iterazioni svolte ne basta una per tutti i thread

I single è un costrutto di worksharing
I Il primo thread che lo raggiunge esegue il blocco associato
I Gli altri lo saltano



Parallelizzazione incrementale: Fortran

Fortran

!$omp parallel
do while (var > tol .and. iter <= maxIter)

!$omp single
iter = iter + 1;
var = 0.d0

!$omp end single
!$omp do reduction(max:var)

do j=1, n
do i=1, n
Tnew(i,j)=0.25d0*(T(i-1,j)+T(i+1,j)+T(i,j-1)+T(i,j+1))
var = max(var,abs( Tnew(i,j) - T(i,j) ))

end do
end do

!$omp end do
!$omp single

Tmp =>T; T =>Tnew; Tnew => Tmp;
if (mod(iter,100)==0) write( *,* ) ’ iter, var:’,iter,var

!$omp end single
end do

!$omp end parallel

Parallelizzazione incrementale: C
C

#pragma omp parallel
{

while(var > tol && iter <= maxIter) {
#pragma omp single

{
++iter;
var = 0.0;
}
double pvar = 0.0;

#pragma omp for private(j)
...

#pragma omp critical
if( pvar > var) var = pvar;

#pragma omp single
{

Tmp=T; T=Tnew; Tnew=Tmp;
if (iter%100 == 0)
printf("iter: %8u, var=%12.4lE\n",iter,var);

}
}
}



Barriere
I Problemi:

I Le variabili iter e var non devono essere aggiornate finché
tutti i thread non hanno controllato la condizione nel while

I L’output non deve essere eseguito finché tutti i thread non
hanno aggiornato var

I Barriere esplicite
I Realizzate con il costrutto barrier che non ha un blocco di

codice associato, garantiscono che TUTTO il codice che la
precede è stato eseguito

I un thread giunto ad una barriera deve arrestarsi finché tutti i
thread nel team non la raggiungono

I Barriere implicite
I Alla fine dei costrutti do/for e single
I Ma non del costrutto critical!

I Alcune volte non sono necessarie, e causerebbero
rallentamenti

I Si possono rimuovere con la clausola nowait

Parallelizzazione incrementale: Fortran

Fortran

!$omp parallel
do while (var > tol .and. iter <= maxIter)

!$omp barrier
!$omp single

iter = iter + 1;
var = 0.d0

!$omp end single
!$omp do reduction(max:var)

do j=1, n
do i=1, n
Tnew(i,j)=0.25d0*(T(i-1,j)+T(i+1,j)+T(i,j-1)+T(i,j+1))
var = max(var,abs( Tnew(i,j) - T(i,j) ))

end do
end do

!$omp end do
!$omp single

Tmp =>T; T =>Tnew; Tnew => Tmp;
if (mod(iter,100)==0) write( *,* ) ’ iter, var:’,iter,var

!$omp end single nowait
end do

!$omp end parallel



Parallelizzazione incrementale: C
C

#pragma omp parallel
{

while(var > tol && iter <= maxIter) {
#pragma omp barrier
#pragma omp single

{
++iter;
var = 0.0;
}
double pvar = 0.0;

#pragma omp for nowait private(j)
...

#pragma omp critical
if( pvar > var) var = pvar;

#pragma omp barrier
#pragma omp single nowait

{
Tmp=T; T=Tnew; Tnew=Tmp;
if (iter%100 == 0)
printf("iter: %8u, var=%12.4lE\n",iter,var);

}
}
}

Il costrutto workshare

I Il blocco di istruzioni racchiuso nel costrutto workshare viene
diviso in unità di lavoro che sono poi assegnate ai thread

I È supportato solo in Fortran allo scopo di parallelizzare l’array
syntax

I Il compilatore provvede a tradurre l’array syntax in loop

I Svantaggi:
I Molte implementazioni sono inefficenti
I Le possibilità di controllo sono limitate

I Nel caso la performance del costrutto workshare non sia
soddisfacente conviene tradurre a mano l’array syntax nei cari,
vecchi loop



Equazione di Laplace, l’ultimo sforzo!

Fortran

!$omp parallel workshare
T(0:n,0:n) = 0.d0;
T(n+1,1:n) = (/ (i, i=1,n) /) * (top / (n+1))
T(1:n,n+1) = (/ (i, i=1,n) /) * (top / (n+1))

!$omp end parallel workshare

Tnew = T

!$omp parallel
do while (var > tol .and. iter <= maxIter)

!$omp barrier
!$omp single

iter = iter + 1;
var = 0.d0

!$omp end single
...

end do
!$omp end parallel

Equazione di Laplace, l’ultimo sforzo!

Fortran

!$omp parallel
!$omp workshare

T(0:n,0:n) = 0.d0;
T(n+1,1:n) = (/ (i, i=1,n) /) * (top / (n+1))
T(1:n,n+1) = (/ (i, i=1,n) /) * (top / (n+1))

!$omp end workshare
!$omp single

Tnew = T
!$omp end single nowait

do while (var > tol .and. iter <= maxIter)
!$omp barrier
!$omp single

iter = iter + 1;
var = 0.d0

!$omp end single
...

end do
!$omp end parallel

Calcolo del potenziale

I N cariche puntiformi positive
identiche

I Potenziale coulombiano:

V =

{ 1
r se r ≤ cutoff
0 se r > cutoff

I Condizioni periodiche

Calcolo delle forze e dell’energia

Fortran
program nbody

implicit none
real(kind(1.d0)) :: pos(3,DIM), forces(3,DIM), f(3), ene
real(kind(1.d0)) :: rij(3), d2, d, cut2=1000
integer :: i, j, k
read( *,* ) pos !legge le posizioni dei pianeti
forces = 0.d0
ene = 0.d0
do i = 1, DIM

do j = i+1, DIM
rij(:) = pos(:,i) - pos(:,j)
d2 = 0.d0
do k = 1, 3

d2 = d2 + rij(k)**2
end do
if (d2 .le. cut2) then

d = sqrt(d2)
f(:) = - 1.d0 / d**3 * rij(:)
forces(:,i) = forces(:,i) + f(:)
forces(:,j) = forces(:,j) - f(:)
ene = ene + (-1.d0/d)

end if
end do

end do
write ( *,fmt=’(e20.10)’ ) ene
write ( *,fmt=’(i5,1x,3e20.10)’ ) (i, forces(:, i), i =1, DIM)

end program nbody



Calcolo delle forze e dell’energia
C
#include <stdio.h>
#include <math.h>
#include <stdlib.h>
int main() {

char fn[FILENAME_MAX];
FILE *fin;
double ( *pos )[3], ( *forces )[3];
double rij[3], d, d2, d3, ene, cut2=1000;
unsigned i, j, k, nbodies=DIM;
/* read input file */
ene = 0.0;
for(i=0; i<nbodies; ++i)

for(j=i+1; j<nbodies; ++j) {
d2 = 0.0;
for(k=0; k<3; ++k) {

rij[k] = pos[i][k] - pos[j][k];
d2 += rij[k]*rij[k];

}
if (d2 <= cut2) {

d = sqrt(d2);
d3 = d*d2;
for(k=0; k<3; ++k) {

double f = -rij[k]/d3;
forces[i][k] += f;
forces[j][k] -= f;

}
ene += -1.0/d;

}
}
/* write energy and forces */

return 0;
}

Test seriale

I Compiliamo il codice seriale:
ifort -fast -DDIM=55000 Nbody.F90 -o nbodyf

icc -fast -DDIM=55000 Nbody.c -o nbodyc

I Eseguiamolo e salviamo l’output:
nbodyf < positions.xyz.55000 > outf

nbodyc < positions.xyz.55000 > outc



Parallelizzazione
Fortran

!$omp parallel do private(i,j,k,rij,d,d2,f) reduction(+:ene)
do i = 1, DIM

do j = i+1, DIM
rij(:) = pos(:,i) - pos(:,j)
d2 = 0.d0
do k = 1, 3

d2 = d2 + rij(k)**2
end do
if (d2 .le. cut2) then

d = sqrt(d2)
f(:) = - 1.d0 / d**3 * rij(:)
do k=1,3
forces(:,i) = forces(:,i) + f(:)

forces(:,j) = forces(:,j) - f(:)

ene = ene -1.d0/d
end if

end do
end do

Parallelizzazione
C

#pragma omp parallel for private(i,j,k,rij,d,d2,d3) \
reduction(+:ene)

for(i=0; i<nbodies; ++i)
for(j=i+1; j<nbodies; ++j) {

d2 = 0.0;
for(k=0; k<3; ++k) {

rij[k] = pos[i][k] - pos[j][k];
d2 += rij[k]*rij[k];

}
if (d2 <= cut2) {

d = sqrt(d2);
d3 = d*d2;
for(k=0; k<3; ++k) {

double f = -rij[k]/d3;

forces[i][k] += f;

forces[j][k] -= f;
}
ene += -1.0/d;

}
}



Race condition

Fortran

f(:) = - 1.d0 / d**3 * rij(:)
forces(:,i) = forces(:,i)+f(:)
forces(:,j) = forces(:,j)-f(:)

C

for(k=0; k<3; ++k) {
double f = -rij[k]/d3;
forces[i][k] += f;
forces[j][k] -= f; }

I Esperimento ideale
I La carica 1 interagisce con 2
I la carica 2 interagisce con 3
I Iterazione i = 1 al thread 0, iterazione i = 2 al thread 1

Thread 0
1. Legge le posizioni delle cariche 1 e 2

2. Calcola la forza tra la coppia 1 − 2 (f_12)

3. Legge forces(:, 1) = 0 e forces(:, 2) = 0

4. Somma f_12 a forces(:, 1) = 0 e forces(:, 2) = 0

5. forces(:, 1) = −forces(:, 2) = f_12

Thread 1
1. Legge le posizioni delle cariche 2 e 3

2. Calcola la forza tra la coppia 2 − 3 (f_23)

3. Legge forces(:, 2) = 0 e forces(:, 3) = 0

4. Somma f_23 a forces(:, 2) = 0 e forces(:, 3) = 0

5. forces(:, 2) = −forces(:, 3) = f_23

In seriale invece:
forces(:, 2) = −f_12 + f_23

Parallelizzazione: reduction
Fortran

!$omp parallel do private(i,j,k,rij,d,d2,f) &
!$omp reduction(+:ene,forces)

do i = 1, DIM
do j = i+1, DIM

rij(:) = pos(:,i) - pos(:,j)
d2 = 0.d0
do k = 1, 3

d2 = d2 + rij(k)**2
end do
if (d2 .le. cut2) then

d = sqrt(d2)
f(:) = - 1.d0 / d**3 * rij(:)

forces(:,i) = forces(:,i) + f(:)
forces(:,j) = forces(:,j) - f(:)

ene = ene + (-1.d0/d)
end if

end do
end do

!$omp end parallel do



Il costrutto atomic
I reduction(+:forces) crea una copia dell’array

per ogni thread
I L’uso di memoria può essere elevato
I Variabile di ambiente OMP_STACKSIZE per garantire uno stack

sufficiente ad ogni thread

I Il costrutto atomic si applica solo a statement che aggiornano
il valore di una variabile

I Garantisce che nessun altro thread modifichi la variabile tra la
lettura e la scrittura

I NON è un costrutto critical!
I Solo la modifica è serializzata, e solo se due o più thread

accedono allo stesso indirizzo di memoria
I Può dare significativi vantaggi nel tempo di esecuzione rispetto

ad un costrutto critical

I Le istruzioni permesse differiscono tra Fortran e C/C++
I Consultare le specifiche dello standard

Parallelizzazione: atomic
Fortran

!$omp parallel do private(i,j,k,rij,d,d2,f) reduction(+:ene)
do i = 1, DIM

do j = i+1, DIM
rij(:) = pos(:,i) - pos(:,j)
d2 = 0.d0
do k = 1, 3

d2 = d2 + rij(k)**2
end do
if (d2 .le. cut2) then

d = sqrt(d2)
f(:) = - 1.d0 / d**3 * rij(:)
do k=1,3

!$omp atomic
forces(k,i) = forces(k,i) + f(k)

!$omp atomic
forces(k,j) = forces(k,j) - f(k)

end do
ene = ene -1.d0/d

end if
end do

end do



Parallelizzazione: atomic
C

#pragma omp parallel for private(i,j,k,rij,d,d2,d3) \
reduction(+:ene)

for(i=0; i<nbodies; ++i)
for(j=i+1; j<nbodies; ++j) {

d2 = 0.0;
for(k=0; k<3; ++k) {

rij[k] = pos[i][k] - pos[j][k];
d2 += rij[k]*rij[k];

}
if (d2 <= cut2) {

d = sqrt(d2);
d3 = d*d2;
for(k=0; k<3; ++k) {

double f = -rij[k]/d3;
#pragma omp atomic

forces[i][k] += f;
#pragma omp atomic

forces[j][k] -= f;
}
ene += -1.0/d;

}
}

Parallelizzazione: riduzione esplicita

C
int tot_threads;
double ( *gforces )[3];

#pragma omp parallel private(i,j,k,rij,d,d2,d3)
{

#pragma omp single
{ tot_threads = omp_get_num_threads();

gforces = calloc(nbodies*tot_threads, sizeof( *gforces )); }
//...

I In C/C++ non è definita la reduction sugli array
I Perché in C/C++ gli array non sono tipi aritmetici
I Potremmo procedere come con Laplace, ma avremmo una

regione critical contenente un loop
I Come evitarlo?

I Allochiamo un vettore shared globale, puntato da gforces
I Ed ogni thread ne userà una porzione distinta in base

al proprio rank



Parallelizzazione: riduzione esplicita
C
#pragma omp parallel private(i,j,k,rij,d,d2,d3)
{

double ( *pforces )[3];

#pragma omp single
{ tot_threads = omp_get_num_threads();

gforces = calloc(nbodies*tot_threads, sizeof( *gforces )); }

pforces = gforces + nbodies*omp_get_thread_num();

#pragma omp for reduction(+:ene)
for(i=0; i<nbodies; ++i)

for(j=i+1; j<nbodies; ++j) {
d2 = 0.0;

for(k=0; k<3; ++k) {
rij[k] = pos[i][k] - pos[j][k];
d2 += rij[k]*rij[k];

}
if (d2 <= cut2) {

d = sqrt(d2);
d3 = d*d2;
for(k=0; k<3; ++k) {

double f = -rij[k]/d3;
pforces[i][k] += f;
pforces[j][k] -= f;

}
ene += -1.0/d;

}
//...

Parallelizzazione: riduzione esplicita

C

//...

#pragma omp for
for(i=0; i<nbodies; ++i)

for(j=0; j<tot_threads; j++)
for(k=0; k<3; ++k)

forces[i][k] += gforces[i+j*nbodies][k];
}



Sbilanciamento tra i thread

- i

?
j

@
@
@

@
@
@

@
@
@@

Thr. 0 Thr. 1 Thr. 2 Thr. 3

Loop nest triangolare: il
numero di iterazioni in j è
una funzione linearmente
decrescente di i

Il lavoro di ogni thread è
proporzionale all’area
assegnata del triangolo di
iterazioni

Il thread 0 ha sempre più
lavoro degli altri e determina il
tempo di esecuzione

I Per default le iterazioni del loop parallelizzato vengono
suddivise tra gli N thread in N chunk di pari dimensioni

I La clausola schedule(tipo[,chunk_size]) consente di
controllare la distribuzione del lavoro

static scheduling

I Il numero di iterazioni del loop
parallelo viene diviso in chunk
di dimensione costante,
assegnati in round robin ai
thread in ordine di thread_id
crescente

I Di default la dimensione del
chunk è pari a Niter/Nthreads
(salvo arrotondamenti)

I Esempio:
!$omp parallel do &
!$omp schedule(static,3)



dynamic scheduling

I Il numero di iterazioni del loop
parallelo viene diviso in chunk
di dimensione costante ed
ognuno è assegnato ad un
thread non appena questo
termina l’esecuzione del chunk
precedente

I Di default la dimensione del
chunk è pari a 1

I Esempio:
!$omp parallel do &
!$omp schedule(dynamic,1)

guided scheduling

I Il numero di iterazioni è diviso
in chunk di dimensione
decrescente ed ognuno è
assegnato ad un thread non
appena questo termina
l’esecuzione del chunk
precedente

I Di default la dimensione
minima del chunk è pari a 1

I Esempio:
!$omp parallel do &
!$omp schedule(guided,1)



runtime scheduling

I Lo schedule e la dimensione dei chunk possono essere
definite nel momento dell’esecuzione del programma
attraverso la variabile d’ambiente OMP_SCHEDULE

I Esempio:
!$omp parallel do &
!$omp schedule(runtime)

I Lo scheduling potrà essere cambiato senza ricompilare il
programma modificando la variabile d’ambiente
OMP_SCHEDULE, ad esempio:

setenv OMP_SCHEDULE “dynamic,50”

I Utile solo a fini sperimentali durante la parallelizzazione

Parallelizzazione: runtime scheduling
Fortran

!$omp parallel do private(i,j,k,rij,d,d2,f) &
!$omp reduction(+:ene,forces) &
!$omp schedule(runtime)

do i = 1, DIM
do j = i+1, DIM

rij(:) = pos(:,i) - pos(:,j)
d2 = 0.d0
do k = 1, 3

d2 = d2 + rij(k)**2
end do
if (d2 .le. cut2) then

d = sqrt(d2)
f(:) = - 1.d0 / d**3 * rij(:)

forces(:,i) = forces(:,i) + f(:)
forces(:,j) = forces(:,j) - f(:)

ene = ene + (-1.d0/d)
end if

end do
end do



Parallelizzazione: runtime scheduling
C
#pragma omp parallel private(i,j,k,rij,d,d2,d3)
{

double ( *pforces )[3];

#pragma omp single
{ tot_threads = omp_get_num_threads();

gforces = calloc(nbodies*tot_threads, sizeof( *gforces )); }

pforces = gforces + nbodies*omp_get_thread_num();
#pragma omp for reduction(+:ene) schedule(runtime)

for(i=0; i<nbodies; ++i)
for(j=i+1; j<nbodies; ++j) {

d2 = 0.0;
for(k=0; k<3; ++k) {

rij[k] = pos[i][k] - pos[j][k];
d2 += rij[k]*rij[k];

}
if (d2 <= cut2) {

d = sqrt(d2);
d3 = d*d2;
for(k=0; k<3; ++k) {

double f = -rij[k]/d3;
pforces[i][k] += f;
pforces[j][k] -= f;

}
ene += -1.0/d;

}

//...

Test parallelo

I Misurare i tempi delle tre versioni (seriale, parallela con
atomic e parallela con reduction)

I Calcolare le performance (speedup ed efficienza) delle versioni
parallele rispetto a quella seriale

I Come variano le prestazioni al variare del tipo di
scheduling/chunking?

I I risultati delle differenti versioni sono uguali?
I Dovrebbero essere uguali?

Il costrutto sections

Fortran
!$omp sections
!$omp section
! codice
!$omp section
! altro codice
!...
!$omp end sections

C
#pragma omp sections
{

#pragma omp section
// codice

#pragma omp section
// altro codice

// ...
}

I È un costrutto di worksharing per distribuire blocchi di codice
che possono essere eseguiti in parallelo

I Ogni thread riceve una section
I Quando un thread ha finito di eseguire una section, ne riceve

un’altra
I Se non ce ne sono altre da eseguire, aspetta che gli altri

finiscano

Il costrutto master

Fortran
!$omp master
write ( *,* ) ’Residual:’, res

!$omp end master

C
#pragma omp master
printf("Residual: %lf\n", res);

I Solo il Master Thread esegue il blocco di codice associato
I Gli altri lo ignorano senza aspettare



Altre clausole per i dati

I firstprivate(list)
I Come private(list)
I Ma inizializza ogni copia col valore della variabile originale

I copyprivate(list)
I Solo sul costrutto single
I Le variabili elencate in list devono essere già private
I Alla fine del costrutto copia i valori nelle copie locali al thread

che lo ha eseguito nelle copie private di tutti gli altri thread
I Di fatto è un’operazione di broadcast da un thread a tutto il

team

Orphaning
I Supponiamo di avere una funzione o subroutine
I Che vogliamo chiamare sia da regioni seriali che da regioni

parallele del codice
I E che nelle regioni parallele le elaborazioni debbano essere

distribuite tra i thread

I Non è necessario scrivere una versione parallela ed una
seriale

I Possiamo avere un’unica versione ed inserirvi i costrutti di
worksharing opportuni

I Un costrutto di worksharing non incluso lessicalmente in un
costrutto parallel:

I è detto orfano
I viene eseguito serialmente se chiamato da una regione seriale
I ma le variabili dichiarate private saranno comunque

privatizzate



La direttiva threadprivate

I È una direttiva dichiarativa
I Si applica:

I In Fortran, a blocchi common o a variabili di modulo
I In C/C++ a variabili a file-scope, namespace-scope o static

I Deve seguire la dichiarazione della variabile, nella stessa unità
di programma

I Permette di creare una copia privata per ciascun thread di
variabili globali al programma

I Fondamentale per rendere thread safe funzioni di librerie
I E per parallelizzare codici suddivisi in più file sorgente che

accedono a variabili globali

I È possibile inizializzarle con i valori della copia del Master
Thread tramite la clausola copyin

Decomposizione LU

I Un caso molto comune in algoritmi
per l’algebra lineare

I Le frecce indicano dipendenze
nell’ordine di esecuzione

I Il grado di parallelismo varia nel
corso del calcolo

I Una gestione rigida riduce l’efficienza
di un fattore 2



Un caso comune

Fortran
do while(associated(p))

call process(p)
p => p%next

end do

C
while(p) {

process(p);
p = p->next;

}

I È un pointer chasing loop
I Il numero di iterazioni non è noto all’inizio del loop
I Ogni iterazione dipende dalla precedente per la

determinazione del nuovo p

I Non è parallelizzabile tramite un costrutto do/for

Parallelismo irregolare

I Macchine sempre più potenti
I Problemi sempre più grandi e complessi
I Algoritmi e strutture dati sempre più complicati

I Algoritmi su grafi
I Algoritmi di ricerca su sequenze di simboli
I Ottimizzazioni combinatorie
I Simulazioni agent-based
I Discretizzazioni su griglie multiblocco complesse
I Embedded Grids
I Adaptive Mesh Refinement
I Simulazioni di circuiti elettronici e chip complessi
I Simulazioni di veicoli spaziali

I Sempre più applicazioni sono caratterizzate da parallelismo
irregolare, che cambia durante l’esecuzione in maniera non
facilmente prevedibile



Il costrutto task

I Ogni volta che un thread incontra un costrutto task
I genera un task
I corrispondente all’esecuzione del blocco di codice associato
I e dotato di un suo data environment

I Un task può essere eseguito immediatamente
I Oppure la sua esecuzione può essere differita, o affidata ad un

altro thread
I Tutti i thread arrivati ad una barriera eseguono i task ancora in

coda finché non sono tutti completati

Trattare una lista con i task

Fortran
!$omp parallel
!$omp single

do while(associated(p))
!$omp task firstprivate(p)

call process(p)
!$omp end task
p => p%next

end do
!$omp end single
!$omp end parallel

Fortran
#pragma omp parallel
{
#pragma omp single

while(p) {
#pragma omp task firstprivate(p)

process(p);
p = p->next;

}
}

I Un thread genera task per tutti
I Quando finisce di generarli arriva alla barriera implicita e

comincia anche lui ad eseguirli

I Ma che valore assume p in ogni task?
I Quello che aveva al momento della creazione del task



Load balancing con i task

Fortran
!$omp parallel private(p)

!$omp do
do i=1, numlists

p => listheads(i)%list
do while(associated(p))

!$omp task
call process(p)

!$omp end task
p => p%next

end do
end do
!$omp end do

!$omp end parallel

C
#pragma omp parallel private (p)
{

#pragma omp for
for (i=0; i<num_lists; i++) {

p = listheads[i];
while(p) {

#pragma omp task
process(p);

p=next(p);
}

}
}

I Un numero dato di liste, di lunghezza variabile
I Assegnarne una per thread provocherebbe uno

sbilanciamento
I Con i task è possibile bilanciare il carico di lavoro
I Per default, variabili private diventano firstprivate in un

costrutto task

Strutture ad albero binario

Fortran
recursive subroutine preorder(p)

type(node), pointer :: p
call process(p%data)
if (associated(p%left))

!$omp task
call preorder(p%left)

!$omp end task
end if
if (associated(p%right))

!$omp task
call preorder(p%right)

!$omp end task
end if

end subroutine

C
void preorder(node *p) {

process(p->data);
if (p->left)

#pragma omp task
preorder(p->left);

if (p->right)
#pragma omp task

preorder(p->right);
}

I Un task può generarne altri
I Anche in modo ricorsivo
I L’albero è attraversato in preorder : ogni nodo è elaborato

prima di procedere ai figli
I I dummy argument diventano firstprivate in un costrutto
task orfano



Albero binario: attraversamento postorder

Fortran
recursive subroutine postorder(p)

type(node), pointer :: p
if (associated(p%left))

!$omp task
call postorder(p%left)

!$omp end task
end if
if (associated(p%right))

!$omp task
call postorder(p%right)

!$omp end task
end if

!$omp taskwait

call process(p%data)
end subroutine

C
void postorder(node *p) {

if (p->left)
#pragma omp task

postorder(p->left);
if (p->right)

#pragma omp task
postorder(p->right);

#pragma omp taskwait

process(p->data);
}

I Nell’attraversamento in postorder ogni nodo è elaborato dopo
l’elaborazione dei figli

I Il costrutto taskwait:
I sospende l’esecuzione di un task
I finché l’esecuzione dei task che lui ha generato non è stata

completata

Come andare avanti

I Questo corso introduttivo consente di leggere e scrivere in
OpenMP

I Ci sono altre funzionalità che non abbiamo descritto
I Ci sono modi più o meno efficienti di fare le cose
I Ci sono regole e restrizioni che dovrete conoscere in futuro

I Dove cercare approfondimenti?
I Nel documento di specifiche di OpenMP, scaricabile dal sito
http://www.openmp.org

I Nei forum del sito http://www.openmp.org
I Nel libro Using OpenMP, di B. Chapman, G. Jost, R. Van der

Pas, MIT Press
I O, sul parallelismo in generale, nel libro Patterns for Parallel

Programming, di T. Mattson, B. Sanders, B. Massingill,
Addison-Wesley

